IoT-Based Smart Vehicle Safety System

1. Project Summary

The primary objective of this project is to prevent road accidents and ensure vehicle safety. It is an IoT (Internet of Things) based system that monitors the real-time conditions of a vehicle and sends data to the cloud (Blynk App).

- **Accident Prevention & Detection:** An ultrasonic sensor detects obstacles in front of the vehicle, and an MPU6050 sensor detects if the vehicle flips over (accident detection).
- **Environment & Driver Safety:** An MQ-4 sensor detects gas leaks inside the vehicle, and an MQ-3 sensor checks the driver's alcohol level (drunken state) to prevent the vehicle from starting or trigger an alert. A fire sensor generates instant alerts if a fire breaks out in the engine or vehicle.
- **Immediate Action:** During any emergency, an on-board buzzer and LCD display are activated, and live notifications are sent directly to mobile phones anywhere in the world via the Blynk app.

2. Pin Connection Chart

All components will be connected to the NodeMCU and breadboard according to the following chart:

Component	Sensor Pin	NodeMCU Pin (ESP8266)	Power Line
Ultrasonic Sensor (HC-SR04)	TRIG ECHO	D5 D6	VCC → 5V GND → GND
MPU6050 (Gyro/Accel)	SDA SCL	D2 D1	VCC → 3.3V GND → GND
LCD Display (I2C Module)	SDA SCL	D2 D1	VCC → 5V GND → GND
MQ-4 Gas Sensor	AO (Analog Output)	A0	VCC → 5V GND → GND
Flame/Fire Sensor	DO (Digital Output)	D3 (or D0 if boot issue occurs)	VCC → 5V GND → GND
MQ-3 Alcohol Sensor	DO (Digital Output)	D4 (or RX if boot issue occurs)	VCC → 5V GND → GND
Output LED	Positive (+)	D7	Negative (-) → GND
Output Buzzer	Positive (+)	D8	Negative (-) → GND

3. Complete Blynk Code:

```
#define BLYNK_TEMPLATE_ID "TMPL6g-ebZ-CU"
#define BLYNK_TEMPLATE_NAME "VehicleSafety"
#define BLYNK_AUTH_TOKEN "9781scGwi8VpFS3O1F19UslwE3LGwqf9"

#include <ESP8266WiFi.h>
#include <BlynkSimpleEsp8266.h>
#include <Wire.h>
#include <MPU6050.h>
#include <LiquidCrystal_I2C.h>

char ssid[] = "XXXX";
char pass[] = "XXXX";

BlynkTimer timer;
MPU6050 mpu;
LiquidCrystal_I2C lcd(0x27, 16, 2);

#define TRIG D5
#define ECHO D6
#define LED D7
#define BUZZ D8
#define MQ A0


#define FLAME_PIN D3
#define ALCOHOL_PIN D4

long getDistance() {
 digitalWrite(TRIG, LOW);
 delayMicroseconds(2);
 digitalWrite(TRIG, HIGH);
 delayMicroseconds(10);
 digitalWrite(TRIG, LOW);
 long duration = pulseIn(ECHO, HIGH, 25000);
 if (duration == 0) return 400;
 return duration * 0.034 / 2;
```

```
}
void sendData() {
 long dist = getDistance();
 int gas = analogRead(MQ);

 int fireDetected = digitalRead(FLAME_PIN);
 int alcoholDetected = digitalRead(ALCOHOL_PIN);

 int16_t ax, ay, az, gx, gy, gz;
 mpu.getMotion6(&ax, &ay, &az, &gx, &gy, &gz);
 float tilt = abs(ax / 16384.0 * 90);

 String alert = "SAFE";
 bool hazard = false;

 if (dist < 15) {
  alert = "OBSTACLE";
  hazard = true;
 }
 else if (gas > 900) {
  alert = "GAS LEAK";
  hazard = true;
 }
 else if (fireDetected == LOW) {
  alert = "FIRE ALERT";
  hazard = true;
 }
 else if (alcoholDetected == LOW) {
  alert = "ALCOHOL DET";
  hazard = true;
 }
 else if (tilt > 45) {
  alert = "ACCIDENT";
  hazard = true;
 }

 if (hazard) {
  digitalWrite(LED, HIGH);
```

```
  digitalWrite(BUZZ, HIGH);
 } else {
  digitalWrite(LED, LOW);
  digitalWrite(BUZZ, LOW);
 }

 lcd.clear();
 if (hazard) {
  lcd.setCursor(0, 0);
  lcd.print("HAZARD DETECT!");
  lcd.setCursor(0, 1);
  lcd.print("STATUS: " + alert);
 } else {
  lcd.setCursor(0, 0);
  lcd.print("D:" + String(dist) + "cm G:" + String(gas));
  lcd.setCursor(0, 1);
  lcd.print("T:" + String((int)tilt) + " F:" + String(fireDetected == LOW ? "Y" : "N"));
 }

 Blynk.virtualWrite(V0, dist);
 Blynk.virtualWrite(V1, gas);
 Blynk.virtualWrite(V2, tilt);
 Blynk.virtualWrite(V3, alert);
 Blynk.virtualWrite(V4, fireDetected == LOW ? 1 : 0);
 Blynk.virtualWrite(V5, alcoholDetected == LOW ? 1 : 0);
}

void setup() {
 Serial.begin(115200);
 pinMode(TRIG, OUTPUT);
 pinMode(ECHO, INPUT);
 pinMode(LED, OUTPUT);
 pinMode(BUZZ, OUTPUT);

 pinMode(FLAME_PIN, INPUT);
 pinMode(ALCOHOL_PIN, INPUT);

 digitalWrite(LED, LOW);
```

```
 digitalWrite(BUZZ, LOW);
 Wire.begin(D2, D1);
 mpu.initialize();
 lcd.init();
 lcd.backlight();

 lcd.setCursor(0, 0);
 lcd.print("Vehicle Safety");
 lcd.setCursor(0, 1);
 lcd.print("System Ready...");

 WiFi.begin(ssid, pass);
 Blynk.config(BLYNK_AUTH_TOKEN);
 timer.setInterval(1000L, sendData);
}

void loop() {
 if (WiFi.status() == WL_CONNECTED) {
  Blynk.run();
 }
 timer.run();
}
```

---

4. Future Scope

- **Automatic Engine Cut-off:** By using a Relay Module, the vehicle's engine can be automatically turned off or locked if the driver is found intoxicated or if a fire breaks out.
- **GPS & GSM Integration:** By integrating a GPS module, the precise live location of the vehicle can be sent instantly via SMS with a Google Maps link to emergency services (like 999) or family members as soon as an accident happens.
- **AI Driver Monitoring:** Mounting a camera on the dashboard can track the driver's eye movements via image processing to detect drowsiness or sleeping, triggering an immediate alert.
- **Smart Blackbox Data Logger:** Adding an SD card module will allow the system to continuously record all sensor data from the 10 seconds leading up to an accident, serving as an investigative tool similar to an airplane or automotive black box.



Fig: Block Diagram



Fig: Connection diagram